<?xml version="1.0" encoding="UTF-8"?>

<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
	
	<xsl:include href="../../admin.xsl" />
	<xsl:include href="../../cp/path.xsl" />
	
	<!-- Запускаем рендер страницы -->
	<xsl:template match="/data">
		<html>
		
		<head>
			<title>
				<xsl:value-of select="$pageTitle" />
			</title>
			<xsl:call-template name="meta-head" />
			<link rel="stylesheet" href="{$baseURI}/css/admin/system_types.css" type="text/css" media="all" />
			<script type="text/javascript" src="{$baseURI}/js/admin/system.js" />
		</head>
		<body>
			<xsl:call-template name="admin-header" />
			
			<div id="blocks">
				<div id="blocks_inner" />
			</div>
			
			<div id="content">
				<div id="content_inner">
					<xsl:apply-templates select="response" />
				</div>
			</div>
			
			<xsl:call-template name="footer" />
		</body>
		</html>
	</xsl:template>
	
	<xsl:template match="response">
		<div id="path">
			<xsl:apply-templates select="document('xmodule://cp/getPath/0;system;types')/data/response" mode="cp-path" />
		</div>
		
		<h1>Типы данных</h1>
		
		<xsl:apply-templates select="message" />
		
		<p>
			<a class="add" href="{$baseURI}/admin/system/add_type">Создать новый тип</a>
			<xsl:text disable-output-escaping="yes">&amp;nbsp;&amp;nbsp;&amp;nbsp;&amp;nbsp;&amp;nbsp;</xsl:text>
			<a class="import" href="{$baseURI}/admin/system/types" onclick="datatypesImport();return false;">Импортировать</a>
		</p>
		
		<form action="{$baseURI}/admin/system/export_types" method="post">		
			<table class="standart canHightlightRows" cellspacing="0" cellpadding="0">
				<tbody>
					<tr>
						<th />
						<th>Название</th>
						<th>Алиас</th>
						<th>Таблица</th>
						<th>Опции</th>
					</tr>
					
					<xsl:apply-templates select="object[@type='explay_type']" />
				</tbody>
			</table>
			
			<p>
				<input type="submit" value="Экспортировать" />
			</p>
		</form>
	</xsl:template>
	
	<xsl:template match="object[@type='explay_type']">
		<tr>
			<td class="checkboxTd">
				<input type="checkbox" name="types[]" value="{@id}" />
			</td>
			<td>
				<a href="{$baseURI}/admin/system/fields/{@id}" class="mainlink" title="Поля">
					<xsl:value-of select="property[@name='title']/value" />
				</a>
			</td>
			<td>
				<xsl:value-of select="property[@name='name']/value" />
			</td>
			<td>
				<xsl:value-of select="property[@name='table_name']/value" />
			</td>
			<td>
				<a href="{$baseURI}/admin/system/edit_type/{@id}">редактировать</a>
				<xsl:if test="property[@name='locked']/value = 0">
					<xsl:text> ∙ </xsl:text>
					<a href="{$baseURI}/admin/system/delete_type/{@id}?scode={$scode}" onclick="if(!confirm('Вы действительно хотите удали этот тип данных? После удаления все данные этого и зависимых типов будут потеряны!'))return false;">удалить</a>
				</xsl:if>
			</td>
		</tr>
	</xsl:template>
	
</xsl:stylesheet>